Survival Models in SAS
Part 10: Competing Risks

November 19, 2008

Charlie Hallahan

Chapter 6: Competing Risks

These talks are based on the book "**Survival Analysis Using the SAS System: A Practical Guide**" (1995) by Paul Allison.

The book is part of the SAS Books-by-Users series and can be found at http://www.sas.com/apps/pubscat/bookdetails.jsp?catid=1&pc=55233



Chapter 6: Competing Risks

This series of talks will cover

Chapter 1: Introduction

Chapter 2: Basic Concepts of Survival Analysis

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG

Chapter 5: Estimating Cox Regression Models with PROC PHREG

Chapter 6: Competing Risks

Chapter 6: Competing Risks

Topics in Chapter 6:

Introduction
Type-Specific hazards
Time in Power for Leaders of Countries: Example
Estimates and Tests without Covariates
Covariate Effects via Cox Models
Accelerated Failure Time Models
An Alternative Approach to Multiple Event Types
Conclusion

Chapter 6: Introduction

In many situations, there can be more than one reason for failure.

For example, in a study evaluating a new heart disease medicine, failure could occur due to heart failure (the failure of interest), cancer, accident, etc.

In a study of job terminations, failure could be due to being fired or quitting.

In a study of inmate recidivism, financial aid to released convicts will more plausibly reduce arrests for theft or burglary than for rape or assault.

In all these situations of competing risks, the occurrence of one type of event precludes the subject from experiencing a different event, and the subject is removed from the risk set for the other events.

The same **SAS** procedures, **LIFETEST, LIFEREG,** and **PHREG**, are all still used for the case of competing risks, just with different syntax.

Chapter 6: Type-Specific Hazards

Consider an example where there are five type of deaths: heart disease, cancer, stroke, accident, and other.

Let T_i be a random variable denoting the time of death for person i and J_i a random variable denoting the type of death that occured to person i.

A **type-specific hazard** is defined as: $h_{ij}(t) = \lim_{\Delta t \to 0} \frac{\Pr\left(t \le T_i < t + \Delta t, J_i = j \mid T_i \ge t\right)}{\Delta t}$

An overall hazard is defined as $h_i(t) = \sum_{j=1}^{5} h_{ij}(t)$.

Chapter 6: Type-Specific Hazards

Type-specific survival functions can then be defined as: $S_j(t) = \exp\left\{-\int_0^t h_j(u)du\right\}$.

Interpretation of the $S_j(t)$ becomes an issue in terms of associating the survival function with a random variable.

If we define the unobserved random variable T_{ij} as the time at which the jth event type either occured or *would have occurred if other event types had not preceded it*.

For example, we assume that a person who died of cancer at time T_2 would have later died of heart disease at time T_1, if the cancer death had not occurred first.

Further assuming that the T_{ij}s are independent across event types, we can say that

$$S_{ij}(t) = \Pr\left(T_{ij} \geq t\right).$$

Chapter 6: Type-Specific Hazards

Given that type-specific hazards $h_{ij}(t)$ have now been defined, they can modeled as before as either **proportional hazards** or as **failure time regression models**.

The general proportional hazards model for all five death types would then be:

$$\log h_{ij}(t) = \alpha_j(t) + \boldsymbol{\beta}_j \mathbf{x}_j(t), \qquad j = 1, \ldots, 5$$

Note that one could, if desired, assume a log-normal model for heart disease, a gamma model for cancer, and a proportional hazards model for stroke. This is because each model can be estimated separately for each event type without any loss of efficiency. The reason for this is that the likelihood function of all event types taken together can be factored into separate likelihood functions for each event type.

Chapter 6: Time in Power for Leaders of Countries: Example

The dataset used to illustrate competing risks was created by **Bienen** and **van de Walle (1991).**

The data is for all countries world-wide over the past 100 years or so (from 1991).

For each leader, the following variables are in the dataset:

YEARS - # of years in power. Less than one year coded as 0
LOST - 0=still in power in 1987; 1=exit by constitutional means; 2=death by natural causes; and 3=nonconstitutional exit
MANNER - How the leader reached power: 0=constitutional means; 1=nonconstitutional means
START - Year of entry into power
MILITARY - Background of the leader: 1=military; 0=civilian
AGE - Age of leader in years at entry into power
CONFLICT - Level of ethnic conflict: 1=medium or high; 0=low
LOGINC - Log of per capita GNP in 1973 dollars

Chapter 6: Time in Power for Leaders of Countries: Example

GROWTH - Average annual rate of per capita GNP growth between 1965-1983.
POP - Population, in millions (year not indicated)
LAND - Land area in thousands of square kilometers
LITERACY - Literacy rate (year not included)
REGION - 0=Middle East; 1=Africa; 2=Asia; 3=Latin America; 4=North America, Europe, and Australia

The dataset used by Allison restricts the leadership spells to:

- countries outside of Europe, North America, and Australia
- spells that began in 1960 or later
- only the first leadership spell for those leaders with multiple spells.

This leaves a total of 472 spells, of which 115 were still in progress at the time observations were terminated in 1987.

Chapter 6: Time in Power for Leaders of Countries: Example

Of the remaining spells:

27 ended when the leader died of natural causes

165 were terminated by constitutional procedures

165 were terminated by nonconstitutional means.

Chapter 6: Estimates and Tests Without Covariates

With multiple failure events, a basic question is asking whether or not each type of event can have the same hazard function, i.e., is $h_j(t) = h(t)$ for all j.

The failure rates for the leaders dataset, 27 failures for natural causes and 165 for each of the other two types, suggest a difference in the hazard functions.

Equal hazards would imply a common failure rate or expected frequencies of failure.

Under such a null, all three should have a failure rate of (472 - 115)/3 = 119. The Pearson chi-square for such a test is 88.9 with df = 2, leading to a strong rejection of equal frequencies.

Chapter 6: Estimates and Tests Without Covariates

Even if the frequencies are different, the hazards may still be **proportional** in the sense that $h_j(t) = \omega_j h(t)$ for $j = 1, \ldots, 3$.

This could be examined graphically via **PROC LIFETEST** by estimating the **log - log survivor** functions for each of the three event types.

If the hazards are proportional, the log-log survivor functions should be parallel.

Chapter 6: Estimates and Tests Without Covariates

```
proc lifetest data=survival.leaders notable outsurv=a;
   time years*lost(0,2,3);
run;
proc lifetest data=survival.leaders notable outsurv=b;
   time years*lost(0,1,3);
run;
proc lifetest data=survival.leaders notable outsurv=c;
   time years*lost(0,1,2);
run;
data combined;
   set a b c;
   retain outcome 0;
   if years=0 and survival=1 then outcome=outcome+1;
   lls=log(-log(survival));
run;
title "Log-Log Survival Functions for Leaders Data";
proc gplot data=combined;
   where _censor_=0;
   symbol1 interpol=join color=black v=none line=1;
   symbol2 interpol=join color=black v=none line=2;
   symbol3 interpol=join color=black v=none line=8;
   plot lls*years=outcome;
run;
```

Chapter 6: Estimates and Tests Without Covariates



Graph points to non-proportionality

Chapter 6: Estimates and Tests Without Covariates

We can also examine smoothed hazard plots using the **`%smooth`** macro.

%***smooth***(data=combined,time=years,width=**4**)



This plot supports even more strongly that the three events do not have proportional hazards.

Note that for types 1 and 3 (exits by constitutional and nonconstitutional means resp), over 80% of the exits occurred before the seventh year, so the estimated hazards for later years may be unreliable.

Chapter 6: Estimates and Tests Without Covariates

Cox and **Oates (1984)** proposed a **parametric test** of the proportional hazards model of the form $h_j(t) = \omega_j h(t)$.

Consider the model: $\log h_j(t) = \alpha_0(t) + \alpha_j + \beta_j t$.

Then $\beta_j = \beta$ for all j says that the proportional hazards hypothesis is satisfied.

The alternative says that the log hazards diverge linearly with time.

For two event types, **Cox** and **Oates** show that the above model reduces to a **logistic regression model** for type of event, with time of the event as an independent variable.

For more than two event types, we need a **multinomial logit model** for event type.

Chapter 6: Estimates and Tests Without Covariates

PROC CATMOD is used to estimate the multinomial model.

```
proc catmod data=survival.leaders;
   where lost ne 0;
   direct years;
   model lost=years / noprofile;
run;
```

Note that only those observations where an event actually occurred are used.

The **`direct`** statement says to treat **`years`** as a continuous variable.

Chapter 6: Estimates and Tests Without Covariates

Maximum Likelihood Analysis of Variance

Source	DF	Chi-Square	Pr > ChiSq
Intercept	2	69.33	<.0001
years	2	17.89	0.0001
Likelihood Ratio	42	68.45	0.0061

Analysis of Maximum Likelihood Estimates

Parameter	Function Number	Estimate	Standard Error	Chi-Square	Pr > ChiSq
Intercept	**1**	**0.0134**	**0.1432**	**0.01**	**0.9253**
	2	**-2.5393**	**0.3140**	**65.40**	**<.0001**
years	**1**	**-0.00391**	**0.0267**	**0.02**	**0.8834**
	2	**0.1394**	**0.0359**	**15.10**	**0.0001**

Chapter 6: Estimates and Tests Without Covariates

To interpret the output on the previous page, the reference event type has been set, by default, to be event type 3 (exit by nonconstitutional means).

α_1 and β_1 represent the contrasts of event type 1 (exit by constitutional means) to event type 3. The insignificance of INTERCEPT 1 (α_1) and YEARS 1 (β_1) says that the hazards for these two event type are proportional.

On the other hand, The strong significance of INTERCEPT 2 (α_2) and YEARS 2 (β_2) says that the hazards for the event types 3 and 2 (exit by natural causes) are not proportional.

In particular, the ratio of the hazards for types 2 and 3 increases about at a rate of about 15% per year (100 ($e^{0.1394}$ -1)).

Chapter 6: Covariate Effects via Cox Models

To test whether or not covariates have the same effect for the different types of events, we fit four models.

The first model represents the null that the coefficients are equal across the three event types.

Each of the three subsequent models are for specific event types, 1,2, and 3.

```
* assume same coefficients for all three event types;
proc phreg data=survival.leaders;
   model years*lost(0)=manner start military age conflict
         loginc growth pop land literacy / ties=efron;
   strata region;
run;
```

Chapter 6: Covariate Effects via Cox Models

```
* fit a model for event type 3 (exit by nonconstitutional means);
proc phreg data=survival.leaders;
   model years*lost(0,1,2)=manner start military age conflict
         loginc growth pop land literacy / ties=efron;
   strata region;
run;

* fit a model for event type 2 (exit by natural causes);
proc phreg data=survival.leaders;
   model years*lost(0,1,3)=manner start military age conflict
         loginc growth pop land literacy / ties=efron;
   strata region;
run;

* fit a model for event type 1 (exit by constitutional means);
proc phreg data=survival.leaders;
   model years*lost(0,2,3)=manner start military age conflict
         loginc growth pop land literacy / ties=efron;
   strata region;
run;
```

Chapter 6: Covariate Effects via Cox Models

Model treating all event types as equivalent.

Model Fit Statistics

Criterion	Without Covariates	With Covariates
-2 LOG L	**2596.320**	**2561.077**

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
manner	**1**	**0.38280**	**0.15520**	**6.0835**	**0.0136**	**1.466**
start	**1**	**-0.01753**	**0.00814**	**4.6325**	**0.0314**	**0.983**
military	1	-0.23122	0.16342	2.0018	0.1571	0.794
age	**1**	**0.02274**	**0.00556**	**16.7331**	**<.0001**	**1.023**
conflict	1	0.12709	0.13131	0.9367	0.3331	1.136
loginc	**1**	**-0.18221**	**0.08259**	**4.8665**	**0.0274**	**0.833**
growth	1	-0.00204	0.02138	0.0091	0.9239	0.998
pop	1	-0.0000632	0.0006373	0.0098	0.9210	1.000
land	1	0.0000109	0.0000480	0.0519	0.8197	1.000
literacy	1	0.0007020	0.00321	0.0478	0.8268	1.001

Four covariates are significant.

Chapter 6: Covariate Effects via Cox Models

The strongest effect is age of entry into power, with each additional year of age associated with a 2.3% increase in the risk of leaving power.

Note that if age at origin is a covariate, its coefficient represents the effect of age as a time-dependent variable.

Also,

- leaders who attained power by nonconstitutional means (**`MANNER=1`**), have a 47% greater risk of leaving power

- leaders in countries with higher per capita GNP (**`LOGINC`**), have a lower risk of exit. Note that this variable being logged, the coefficient represents an **elasticity** with respect to the hazard, i.e. a 1% increase in GNP causes a 0.1822% decrease in the hazard of failure.

- the risk of exit declined by about 2% per year (**`START`**) since 1960.

Chapter 6: Covariate Effects via Cox Models

Model for nonconstitutional exits (event type 3):

Model Fit Statistics

Criterion	Without Covariates	With Covariates
-2 LOG L	**1273.445**	**1202.040**

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
manner	**1**	**0.92816**	**0.22047**	**17.7232**	**<.0001**	**2.530**
start	**1**	**-0.03350**	**0.01221**	**7.5325**	**0.0061**	**0.967**
military	**1**	**-0.41380**	**0.22757**	**3.3063**	**0.0690**	**0.661**
age	1	0.00897	0.00845	1.1286	0.2881	1.009
conflict	**1**	**0.50445**	**0.20335**	**6.1540**	**0.0131**	**1.656**
loginc	**1**	**-0.43338**	**0.14135**	**9.4001**	**0.0022**	**0.648**
growth	1	-0.04858	0.03112	2.4377	0.1185	0.953
pop	1	-0.00103	0.00154	0.4424	0.5059	0.999
land	1	0.0000201	0.0000802	0.0629	0.8020	1.000
literacy	1	-0.00569	0.00452	1.5843	0.2081	0.994

Note now that **age** is no longer significant and **military** and **conflict** are.

Chapter 6: Covariate Effects via Cox Models

Leaders who acquired power by nonconstitutional means are 2.5 times as likely as other leaders to exit by nonconstitutional means.

Income has a stronger effect with a 1% increase in per capita GNP now yielding a 0.435 percent decrease in the risk of nonconstitutional exit.

AGE is no longer significant, but still has a small positive effect.

Leaders in countries with ethnic conflict (CONFLICT=1) have a 66% greater chance of nonconstitutional exit.

Chapter 6: Covariate Effects via Cox Models

Model for constitutional exits (event type 1):

Model Fit Statistics

Criterion	Without Covariates	With Covariates
-2 LOG L	1168.455	1139.407

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
manner	1	-0.30828	0.25750	1.4333	0.2312	0.735
start	1	0.00239	0.01201	0.0397	0.8421	1.002
military	1	-0.01204	0.26053	0.0021	0.9631	0.988
age	**1**	**0.02398**	**0.00862**	**7.7304**	**0.0054**	1.024
conflict	1	-0.02985	0.20372	0.0215	0.8835	0.971
loginc	1	-0.13068	0.11939	1.1981	0.2737	0.877
growth	1	0.03391	0.03453	0.9640	0.3262	1.034
pop	1	0.0004115	0.0008461	0.2366	0.6267	1.000
land	1	-0.0000261	0.0000703	0.1379	0.7104	1.000
literacy	**1**	**0.01369**	**0.00561**	**5.9502**	**0.0147**	**1.014**

Chapter 6: Covariate Effects via Cox Models

For leaders in countries with constitutional exits, there are only two significant effects, **`AGE`** and **`LITERACY`**.

Each 1% increase in the literacy rate is associated with a 1.4% reduction in the risk of a constitutional exit.

Model for exits by natural causes (event type 2):

Model Fit Statistics

Criterion	Without Covariates	With Covariates
-2 LOG L	180.484	155.376

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
manner	1	0.30269	0.70202	0.1859	0.6663	1.353
start	**1**	**-0.05851**	**0.03530**	**2.7475**	**0.0974**	**0.943**
military	1	-0.29905	0.78251	0.1460	0.7023	0.742
age	**1**	**0.07938**	**0.02038**	**15.1764**	**<.0001**	**1.083**
conflict	1	-0.55363	0.50851	1.1854	0.2763	0.575
loginc	1	0.19486	0.28309	0.4738	0.4912	1.215
growth	1	0.09228	0.08538	1.1681	0.2798	1.097
pop	1	0.0009563	0.00220	0.1885	0.6642	1.001
land	1	0.0000347	0.0001797	0.0372	0.8471	1.000
literacy	1	-0.01239	0.01361	0.8291	0.3625	0.988

Chapter 6: Covariate Effects via Cox Models

This time, only two variables, **AGE** and **START**, are significant at the 10% level.

Each one year increase in age is associated with an 8.3% increase in the hazard of natural death.

The previous output shows that the parameter estimates differ quite a bit depending on the type of event.

The **PHREG** output from each model can be used to calculate a likelihood ratio test of the null hypothesis of equal coefficients across event types.

We have the following reported values for *-2 log-likelihood* for the four models:

All types combined	2561.08
Nonconstitutional	1202.04
Constitutional	1139.41
Natural Death	155.38

Chapter 6: Covariate Effects via Cox Models

A chi-square statistic is calculated by subtracting the sum of the last three values from the first resulting in a chi-square value of 64.25. The degrees-of-freedom is 30 – 10 = 20 producing a significant value well beyond the 1% level.

Testing whether the coefficients for event types 1 and 3 (exits by nonconstitutional versus constitutional means) differ produces a chi-square value of 50.55 with d.f. = 10. The p-value is well below 0.0001, so the conclusion is to allow for different coefficients for exits by nonconstitutional versus constitutional means.

We can also construct tests for equality of specific coefficients for different event types.

For example, to test for a significant difference in the effect of **`CONFLICT`** for types 1 and 3, we note that the estimate for type 3 is 0.5044 with a se = 0.2034 and for type 1 an estimate of -0.02986 with se = 0.2037.

Chapter 6: Covariate Effects via Cox Models

A 1-degree-of-freedom Wald statistic for $H_0 : \beta_1 = \beta_2$ is $F = \dfrac{\left(\hat{\beta}_1 - \hat{\beta}_2\right)^2}{\left[s.e.\left(\hat{\beta}_1^2\right)\right]^2 + \left[s.e.\left(\hat{\beta}_2^2\right)\right]^2}$

In our example, we get $\dfrac{\left(0.5044 - (-0.02985)\right)^2}{\left[0.2034\right]^2 + \left[0.2037\right]^2} = 3.44$ which is not significant at the 5% level.

It might seem that we should be including a covariance term in the above denominator since each model was estimated using the same dataset.

However, because of the assumption that the event types are independent, which implies that the overall likelihood function factors into the likelihoods for each event type, we don't need a covariance term above.

Chapter 6: Covariate Effects via Cox Models

The tests for a single covariate can be generalized to more than two event types.

For a given covariate, let $\hat{\beta}_j$ be its parameter estimate for event type j, s_j^2 be the squared, estimated standard error of β_j, and let $X_j^2 = \hat{\beta}_j^2 / s_j^2$ be the reported Wald chi-square statistic for testing $H_0 : \beta_j = 0$.

To test $H_0 :\ \beta_j = 0 \ \ \forall j$ we form the chi-square statistic $Q = \sum_j X_j^2$ with $d.f. = \#$ of event types.

To test $H_0 :\ \beta_j = \beta \ \ \forall j$ we form the chi-square statistic $Q = \sum_j X_j^2 / \sum_j \left(\frac{1}{s_j^2} \right)$ with $d.f. = \#$ of event types - 1.

Chapter 6: Accelerated Failure Time Models

Competing risks models for **failure time models** can be handled the same way as with **Cox models**, i.e., treat all events except the one of interest as being censored.

Since the **AFT models** take logs of the covariates, the problem arises as to how to handle covariates with values of zero.

In the **LEADERS dataset**, the variable **`year`** is set to 0 if the leader was in office for less than one year.

One solution would be to assign an arbitrary small number for each 0.
A better solution is to threat each observation with a value of 0 as being **left censored.**

For the **LEADERS** dataset, the **`DATA`** step below sets up the two variables upper and lower to reflect the desired interval censoring.

Any event other than the one of interest, **`lost = 3`** in the example, is treated as censored.

Chapter 6: Accelerated Failure Time Models

```
data leaders2;
   set survival.leaders;
   lower=years;
   upper=years;
   if years=0 then do;
      lower=.;
      upper=1;
   end;
   * treat events other than lost = 3 as being censored;
   if lost in (0,1,2) then upper=.;
run;

title "An Exponential AFT Model without any covariates";
proc lifereg data=leaders2;
   model (lower,upper)= / d=exponential;
run;
```

Chapter 6: Accelerated Failure Time Models

The author fits a series of **AFT models** for each of the event types 1, 2, and 3. The variable **`region`** is treated as a classification model, and only significant covariates from any of the previous **PHREG models** are included.

The models and their log-likelihoods are given below:

	Nonconstitutional	Constitutional	Natural Death
Exponential	-383.39	-337.30	-87.17
Weibull	-372.51	-336.46	-82.48
Log-normal	-377.04	-338.09	-83.60
Gamma	-372.47	-336.14	(-81.16)
Log-logistic	-374.95	-335.88	-82.78

Note: the results for the Gamma model for Natural Death (`lost=2`) is in parentheses because of this warning message:

WARNING: The relative gradient convergence criterion of 0.0004324105 is greater than the limit of 0.0001. The convergence is questionable.

Chapter 6: Accelerated Failure Time Models

Using the log-likelihood values above, we can perform likelihood ratio tests by calculating the differences of -2 log-likelihood for each model compared to the most general model, the gamma model.

For example, for the **Nonconstitutional event**, we can easily **reject** the **Exponential** and **Log-normal models**, while the simpler **Weibull model** is essentially **equivalent** to the **Gamma model**.

Chapter 6: Accelerated Failure Time Models

Analysis of Maximum Likelihood Parameter Estimates

Parameter		DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept		1	-40.6136	17.1902	-74.3057	-6.9215	5.58	0.0181
manner		1	-1.3806	0.3143	-1.9965	-0.7646	19.30	<.0001
age		1	-0.0124	0.0114	-0.0348	0.0100	1.18	0.2775
start		1	0.0410	0.0178	0.0062	0.0758	5.33	0.0210
military		1	0.6510	0.3135	0.0366	1.2655	4.31	0.0378
conflict		1	-0.7199	0.2843	-1.2771	-0.1626	6.41	0.0113
loginc		1	0.6750	0.2073	0.2688	1.0812	10.61	0.0011
literacy		1	0.0072	0.0063	-0.0052	0.0196	1.28	0.2572
region	0	1	0.9037	0.4457	0.0301	1.7773	4.11	0.0426
region	1	1	1.3636	0.3926	0.5941	2.1332	12.06	0.0005
region	2	1	2.0362	0.4692	1.1166	2.9558	18.83	<.0001
region	3	0	0.0000	.	.	.	.	.
Scale		**1**	**1.4064**	0.1121	1.2030	1.6441		
Weibull Shape		1	0.7111	0.0567	0.6082	0.8313		

Using the transformation `1/1.4064 – 1 = -0.29` to get the coefficient of log *t* in the **equivalent proportional hazards model**, we see that the hazard of a nonconstitutional exit *decreases* with time since entry into power.

Chapter 6: Accelerated Failure Time Models

Recall that region 3 is Latin America, the reference category in this specification.

The results show that the expected time until a **nonconstitutional exit** is more than seven times greater (exp(2.036)) in Asia (`region=2`) than it is in Latin America, and it nearly four times greater (exp(1.36)) in Africa (`region=1`).

There is a different story for **constitutional exits** as the Table shows that all models have similar log likelihoods.

The **constant hazard Exponential model** is not significantly worse than the more complicated **Weibull** and **Gamma** models.

In the interest of parsimony, the exponential model is estimated for the constitutional exits (**`lost = 1`**).

Chapter 6: Accelerated Failure Time Models

Analysis of Maximum Likelihood Parameter Estimates

Parameter		DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept		1	12.4138	11.7600	-10.6353	35.4630	1.11	0.2912
manner		1	0.3112	0.2624	-0.2030	0.8255	1.41	0.2356
age		**1**	**-0.0317**	**0.0083**	**-0.0480**	**-0.0154**	**14.52**	**0.0001**
start		1	-0.0092	0.0120	-0.0328	0.0144	0.58	0.4453
military		1	0.0284	0.2526	-0.4667	0.5236	0.01	0.9104
conflict		1	0.1573	0.1983	-0.2314	0.5460	0.63	0.4277
loginc		1	0.1366	0.1135	-0.0858	0.3590	1.45	0.2287
literacy		**1**	**-0.0113**	**0.0056**	**-0.0223**	**-0.0003**	**4.09**	**0.0432**
region	0	1	0.5413	0.3326	-0.1106	1.1932	2.65	0.1036
region	**1**	**1**	**1.7032**	**0.3702**	**0.9776**	**2.4288**	**21.17**	**<.0001**
region	**2**	**1**	**0.5336**	**0.2180**	**0.1062**	**0.9609**	**5.99**	**0.0144**
region	3	0	0.0000	.	.	.	.	.
Scale		0	1.0000	0.0000	1.0000	1.0000		
Weibull Shape		0	1.0000	0.0000	1.0000	1.0000		

Lagrange Multiplier Statistics

Parameter	Chi-Square	Pr > ChiSq
Scale	**1.4704**	**0.2253**

Chapter 6: Accelerated Failure Time Models

The only significant covariates for constitutional exits are **`age`** and **literacy**.

Also, we now see that while all regions have longer expected time in power than Latin America, **Africa** now replaces **Asia** as the region with the longest expected time.

Finally, the Lagrange Multiplier test for constant hazard is not rejected.

For exits by **natural causes** (**`lost = 2`**), only the exponential model is demonstrably worse, so the **Weibull model** is estimated below>

Chapter 6: Accelerated Failure Time Models

Analysis of Maximum Likelihood Parameter Estimates

Parameter		DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept		1	-23.1476	20.0574	-62.4595	16.1642	1.33	0.2485
manner		1	-0.2216	0.3909	-0.9878	0.5445	0.32	0.5707
age		**1**	**-0.0451**	**0.0108**	**-0.0662**	**-0.0240**	**17.58**	**<.0001**
start		1	0.0302	0.0209	-0.0107	0.0712	2.09	0.1480
military		1	0.2210	0.4158	-0.5939	1.0360	0.28	0.5950
conflict		1	0.0736	0.2804	-0.4761	0.6232	0.07	0.7930
loginc		1	-0.1514	0.1563	-0.4577	0.1549	0.94	0.3328
literacy		1	0.0030	0.0073	-0.0112	0.0172	0.17	0.6769
region	0	1	0.3597	0.4472	-0.5168	1.2361	0.65	0.4212
region	1	1	0.7672	0.4491	-0.1131	1.6475	2.92	0.0876
region	2	1	0.6378	0.3745	-0.0963	1.3718	2.90	0.0886
region	3	0	0.0000	.	.	.	.	.
Scale		1	0.5994	0.0885	0.4487	0.8007		
Weibull Shape		1	1.6684	0.2465	1.2489	2.2287		

The only significant effect is the expected one, **age**.

Chapter 6: Accelerated Failure Time Models

Suppose we wanted to test for equality of coefficients for various covariates in models for different event types.

For example, suppose we wanted to compare Weibull models for constitutional and nonconstitutional exits from power.

Each model can be written as: $\log h_j(t) = \alpha_j \log t + \beta_{0j} + \beta_{1j} x_1 + \ldots + \beta_{kj} x_k$ where $j = 1$ for constitutional exits and $j = 2$ for nonconstitutional exits.

Thus, $\log \dfrac{\Pr\left(j = 1 \,|\, T = t \right)}{\Pr\left(j = 2 \,|\, T = t \right)} = \left(\alpha_1 - \alpha_2 \right) \log t + \left(\beta_{01} - \beta_{02} \right) + \left(\beta_{11} - \beta_{12} \right) x_1 + \ldots + \left(\beta_{k1} - \beta_{k2} \right) x_k$

Chapter 6: Accelerated Failure Time Models

The above **logit model** can be estimated in **SAS** with several **PROCS: GENMOD, LOGISTIC, PROBIT,** and **CATMOD.**

Since **PROBIT** has a **CLASS** statement, we'll use that PROC.

```
data leaders3;
   set survival.leaders;
   * need to edit zero-values for t;
   lyears=log(years+.5);
run;

proc probit data=leaders3;
   * just keep observations for types 1 and 3 exits;
   where lost=1 or lost=3;
   class lost region;
   model lost=lyears manner age start military conflict loginc
         literacy region / d=logistic;
run;
```

Chapter 6: Accelerated Failure Time Models

Analysis of Maximum Likelihood Parameter Estimates

Parameter		DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept		1	-42.9262	20.7274	-83.5512	-2.3013	4.29	0.0384
lyears		**1**	**0.2749**	**0.1424**	**-0.0043**	**0.5540**	**3.72**	**0.0536**
manner		**1**	**-1.3252**	**0.3801**	**-2.0701**	**-0.5803**	**12.16**	**0.0005**
age		**1**	**0.0277**	**0.0144**	**-0.0005**	**0.0559**	**3.71**	**0.0540**
start		**1**	**0.0407**	**0.0214**	**-0.0013**	**0.0827**	**3.60**	**0.0578**
military		1	0.0698	0.3949	-0.7041	0.8437	0.03	0.8596
conflict		1	-0.2433	0.3425	-0.9147	0.4281	0.50	0.4776
loginc		1	0.0581	0.2502	-0.4323	0.5485	0.05	0.8163
literacy		**1**	**0.0334**	**0.0089**	**0.0158**	**0.0509**	**13.92**	**0.0002**
region	0	1	0.1111	0.5062	-0.8811	1.1033	0.05	0.8263
region	1	1	-0.6684	0.5034	-1.6551	0.3183	1.76	0.1843
region	2	1	0.4737	0.4568	-0.4216	1.3691	1.08	0.2997
region	3	0	0.0000	.	.	.	.	.

We see that there are highly significant differences in the coefficients for **`manner`** and **`literacy`**, and marginally significant differences for **`lyears`**, **`age`**, and **`start`**.

Chapter 6: Accelerated Failure Time Models

Since there is no overall "*F-test*" given for the hypothesis that all the coefficients are zero, i.e., all the coefficients for the two types are exits are the same, we can easily construct a **likelihood ratio** test by estimating the Weibull model without any covariates and constructing a chi-square statistic.

In this case, the resulting statistic is 170.9 with d.f. = 11, which is highly significant.

Chapter 6: An Alternative Approach to Multiple Event Types

The assumption so far has been that each type of competing risk has its own hazard model that determines not only the timing of the event, but also its occurrence.

For example, in comparing being removed from power by constitutional versus nonconstitutional means, it is reasonable to imagine that there is one process that determines whether or not a leader should be removed from power (unpopularity with important sponsors or the general public) and another process, depending on constitutional mechanisms and cultural traditions, determining how the removal takes place.

Taking this approach, one could estimate separate models for each process.

The previous logit model essentially models the second step as to whether removal is by constitutional or nonconstitutional means, where the analysis is restricted to just those leaders experiencing one or the other of these two event types.

Chapter 6: An Alternative Approach to Multiple Event Types

Any of the models previously estimated (Cox or AFT models) would be appropriate for estimating the timing of the event, removal by constitutional or nonconstitutional means.

The model below treats exit by natural causes (`lost = 2`) as a censored observation so that the analysis only deals with the events `lost = 1` or `lost = 3`.

```
proc phreg data=survival.leaders;
   model years*lost(0,2)=manner age start military conflict
      loginc literacy / ties=efron;
   strata region;
run;
```

Chapter 6: An Alternative Approach to Multiple Event Types

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio
manner	**1**	**0.37326**	**0.15841**	**5.5522**	**0.0185**	**1.452**
age	**1**	**0.01705**	**0.00572**	**8.8803**	**0.0029**	**1.017**
start	1	-0.01488	0.00837	3.1588	0.0755	0.985
military	1	-0.20718	0.16368	1.6020	0.2056	0.813
conflict	1	0.16992	0.13564	1.5693	0.2103	1.185
loginc	**1**	**-0.24016**	**0.08843**	**7.3765**	**0.0066**	**0.786**
literacy	1	0.00186	0.00330	0.3175	0.5731	1.002

Conclusions:

- Those who acquired power by nonconstitutional means had a 45% higher risk of losing power (by means other than natural causes)
- Each additional year of age increased the risk of exit by about 1.7%
- A 1% increase in GNP per capita yielded about a .24% decrease in the risk of exit.